Misner Space

Spacetime discussed in Section 2.7, p 81, Carroll

$$ds^2 = \frac{t}{\sqrt{1+t^2}}\, dt^2 - \frac{2}{\sqrt{1+t^2}}\, dt\, dx - \frac{t}{\sqrt{1+t^2}}\, dx^2$$

$0 < x < 1, \ -\infty < t < +\infty$



FIGURE 2.25 A cylindrical spacetime with closed timelike curves. The light cones progressively tilt, such that the domain of dependence of the surface Σ fills the lower part of the spacetime, but comes to an end when the closed timelike curves come into existence.

It is the metric with the above sign that corresponds to the figure in Carroll's book

Null lines

$$ds^2 = 0 \Rightarrow \left(\frac{dx}{dt}\right)^2 + \frac{2}{t}\frac{dx}{dt} - 1 = 0$$

In[ ]:=
```
sol = DSolve[x'[t]^2 + (2/t) x'[t] - 1 == 0, x, t]
```

Out[ ]=
$$\left\{\left\{x \to \text{Function}\left[\{t\}, -\sqrt{1+t^2} + c_1 - \text{Log}\left[-1+\sqrt{1+t^2}\right]\right]\right\}, \left\{x \to \text{Function}\left[\{t\}, \sqrt{1+t^2} + c_1 - \text{Log}\left[1+\sqrt{1+t^2}\right]\right]\right\}\right\}$$

Now, we have to be careful so that the null cones are continuous at the t=0 line.

To do it correctly, we notice that for the 1st solution c_1, when we move in the positive-t direction, then

1. when t < 0, for increasing t, we move towards the right
2. when t >0, for increasing t, we move towards the left. To maintain same forward null cones, we have to move in the direction of decreasing t, therefore reverse the tangent vector

In[ ]:=
```
x1[t_, c_] = ((x[t] /. sol) /. 𝕔1 → c)[[1]]; (* c1 *)
x2[t_, c_] = ((x[t] /. sol) /. 𝕔1 → c)[[2]]; (* c2 *)
{x1[t, c], x2[t, c]}
```

Out[ ]=
$$\left\{c-\sqrt{1+t^2}-\text{Log}\left[-1+\sqrt{1+t^2}\right],\ c+\sqrt{1+t^2}-\text{Log}\left[1+\sqrt{1+t^2}\right]\right\}$$

Null c_1 curves in the t<0 region. They approach the t=0 line, but never reach it. We see that $(2/t)\,\text{dx}/\text{dt} \to 1$ as $t \to 0$

In[ ]:=
```
tmax = 1.5; tmin = 0.021; c0 = 0;
g12m = ParametricPlot[{Mod[x1[t, c0], 1.0], t}, {t, -tmax, -tmin},
   PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}];
g13m = ParametricPlot3D[{ Cos[2 π x1[t, c0]], Sin[2 π x1[t, c0]], t},
  {t, -tmax, -tmin}, PlotRange → All, AxesLabel → {"", "", "t"}];
{g12m, g13m}
```

Out[ ]=


Null c_1 curves in the t<0 region. They approach the t=0 line, but never reach it. We see that $(2/t)\,\text{dx}/\text{dt} \to 1$ as $t \to 0$
Using x1[t,c] we move upwards in t, therefore to the left, which is not compatible with the light cones coming from the t<0 region.

In[∘]:=

```
tmax = 1.5; tmin = 0.021; c0 = 0;
g12p = ParametricPlot[{Mod[x1[t, c0], 1.0], t}, {t, tmin, tmax},
    PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}];
g13p = ParametricPlot3D[{ Cos[2 π x1[t, c0]], Sin[2 π x1[t, c0]], t},
   {t, tmin, tmax}, PlotRange → All, AxesLabel → {"", "", "t"}];
{g12p, g13p}
```

Out[∘]=



Null c_2 curves in the t<0 region. They approach the t=0 line, but never reach it. We see that $(2/t)\,dx/dt \to 1$ as $t \to 0$

In[∘]:=

```
tmax = 5.5; tmin = 0.021; c0 = 0;
g22 = ParametricPlot[{Mod[x2[t, c0], 1.0], t}, {t, -tmax, tmax},
    PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}];
g23 = ParametricPlot3D[{ Cos[2 π x2[t, c0]], Sin[2 π x2[t, c0]], t},
   {t, -tmax, tmax}, PlotRange → All, AxesLabel → {"", "", "t"}];
{g22, g23}
```

Out[∘]=



In[ ]:=

```
tmax = 4.5; tmin = 0.11; c0 = 0;
g12m = ParametricPlot[{Mod[x1[t, c0], 1.0], t}, {t, -tmax, -tmin},
    PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}];
g12p = ParametricPlot[{Mod[x1[t, c0], 1.0], t}, {t,  tmin,  tmax},
    PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}];
g22  = ParametricPlot[{Mod[x2[t, c0], 1.0], t}, {t, -tmax,  tmax},
    PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}];
{Show[g12m, g12p, g22, PlotRange → All, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}],
  Show[g12m, g12p, g22, PlotRange → All,
   AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}, AspectRatio → 1]}
```

Out[ ]=



Now, several different null lines, to see causal structure: Notice here that we are not careful to use the correct parametric equations, so the direction of flowing for t>0 and c_1 curves, is opposite.

In[•]:=

```
tmax = 1.5 ; c0 =.;
range = {{0, 1}, {-1.1 tmax, 1.1 tmax}};
g1 = Table[
    ParametricPlot[{x1[t, c0], t}, {t, -tmax, tmax}, PlotRange → range]
    , {c0, {-4, -1.4, -0.2, 0.5, 1.2, 1.7, 2.2}}];
g2 = Table[
    ParametricPlot[{x2[t, c0], t}, {t, -tmax, tmax}, PlotRange → range]
    , {c0, {-0.5, -0.305, -0.15, 0.0 , 0.2, 0.4, 0.6}}];
Show[{g1, g2}, AspectRatio → 1, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}]
```

Out[•]=



In[ ]:=

```
tmax = 1.5 ; c0 =.;
range = {{0, 1}, {-1.1 tmax, 1.1 tmax}};
g1 = Table[
    ParametricPlot[{x1[t, c0], t}, {t, -tmax, tmax}, PlotRange → range]
    , {c0, {-4, -1.4, -0.2, 0.5, 1.2, 1.7, 2.2}}];
g2 = Table[
    ParametricPlot[{x2[t, c0], t}, {t, -tmax, tmax}, PlotRange → range]
    , {c0, {-0.5, -0.305, -0.15, 0.0 , 0.2, 0.4, 0.6}}];
Show[{g1, g2}, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}]
```

Out[ ]=



Notice the intersections of the curves in Euclidean geometry are at 90^{o}

In[ ]:=
```
tmax = 5.5 ; c0 =.;
range = {{0, 1}, {-1.1 tmax, 1.1 tmax}};
g1 = Table[
    ParametricPlot[{x1[t, c0], t}, {t, -tmax, tmax}, PlotRange → range]
    , {c0, {-4, -1.5, -0.2, 1.0, 2.0, 3, 4, 4.5, 5, 5.5, 6, 6.5, 7.5}}];
g2 = Table[
    ParametricPlot[{x2[t, c0], t}, {t, -tmax, tmax}, PlotRange → range]
    , {c0, {-3.3, -2.7, -2.3, -1.7, -1.3, -0.8, -0.305, 0.0 , 0.3, 0.6}}];
Show[{g1, g2}, AxesOrigin → {0, 0}, AxesLabel → {"x", "t"}]
```

Out[ ]=



Compute tangent vectors.

One has to be careful to maintain the continuity of the causal structure near the t=0 line
For that, the curve c_1 must be transversed in the direction of decreasing t, so that both classes of lines flow to the right.
Therefore, the sign of the tangent vector of c_1, which has parameter t, which is increasing, must be reversed for t>0.

Notice that dx1[t] dx2[t] = -1, therefore the vectors appear normal in Euclidean geometry (of course x-y scale must be the same)

In[ ]:=
```
dx1[t_] = D[x1[t, c], t];
dx2[t_] = D[x2[t, c], t];
dr1[t_] = Which[t < 0, {dx1[t], 1}, t ≥ 0, {-dx1[t], -1}];
dr2[t_] = {dx2[t], 1};
Print[
 "dx1[t]= ", dx1[t], "      dx2[t]= ", dx2[t], "\n",
 "Euclidean dr1·dr2 = dx1 dx2 + 1= ", dx1[t] dx2[t] // Simplify, "+1 = 0"
]
```

dx1[t]= $-\frac{t}{\sqrt{1+t^2}}-\frac{t}{\sqrt{1+t^2}\left(-1+\sqrt{1+t^2}\right)}$ dx2[t]= $\frac{t}{\sqrt{1+t^2}}-\frac{t}{\sqrt{1+t^2}\left(1+\sqrt{1+t^2}\right)}$

Euclidean dr1·dr2 = dx1 dx2 + 1= -1+1 = 0

In[ ]:=
```
tmax = 2.5;
VectorPlot[{dr1[t], dr2[t]}, {x, 0, 1}, {t, -tmax, tmax}, VectorPoints → {5, 10},
 VectorSizes → 1.8, VectorColorFunction → None, AspectRatio → 2,
 VectorStyle → {{Arrowheads[0.02], Red}, {Arrowheads[0.02], Blue}}]
```

Out[ ]=



In[ ]:=

```
tmax = 2.0;
VectorPlot[{dr1[t], dr2[t]}, {x, 0, 1}, {t, -tmax, tmax}, VectorPoints → {5, 10},
 VectorSizes → 1.8, VectorColorFunction → None, AspectRatio → Automatic,
 VectorStyle → {{Arrowheads[0.02], Red}, {Arrowheads[0.02], Blue}}]
```

Out[ ]=



In[ ]:=
```
tmax = 2.0;
VectorPlot[{dr1[t], dr2[t]}, {x, 0, 1}, {t, -tmax, tmax},
 VectorPoints → {5, 10}, VectorSizes → 3, VectorScaling → "Linear",
 VectorColorFunction → None, AspectRatio → Automatic,
 VectorStyle → {{Arrowheads[0.02], Red}, {Arrowheads[0.02], Blue}}]
```

Out[ ]=



Timelike Geodesics

The geodesic equations are:

$$\ddot{t} = \frac{-1}{2\left(1+t^2\right)}\left(-t\,\dot{t}^2 + 2\,\dot{t}\,\dot{x} + t\,\dot{x}^2\right)$$

$$\ddot{x} = \frac{1}{2\left(1+t^2\right)}\left(\left(1+2\,t^2\right)\dot{t}^2 - 2\,t\,\dot{t}\,\dot{x} + \dot{x}^2\right)$$

With initial conditions $x(0) = x_0,\; t(0) = t_0,\; \dot{x}(0) = \dot{x}_0,\; \dot{t}(0) = \dot{t}_0$

But they are not all free due to the constraint:

$$v^\mu v_\mu = -1 \Rightarrow -t\,\dot{t}^2 + 2\,\dot{t}\,\dot{x} + t\,\dot{x}^2 - \sqrt{1+t^2} = 0$$

In[ ]:=
```
dt0 =.; dx0 =.; dt =.; t0 =.; x0 =.;
{initt  = Solve[-t0 dt^2 + 2 dt dx0 + t0 dx0^2 - Sqrt[1 + t0^2] == 0, dt][[1, 1]]
 (* we pick dx0>0 for t0 < 0 *),
 initt2 = Solve[-t0 dt^2 + 2 dt dx0 + t0 dx0^2 - Sqrt[1 + t0^2] == 0, dt][[2, 1]]}
```

Out[ ]=

$$\left\{dt \to \frac{dx0 - \sqrt{dx0^2 + dx0^2\, t0^2 - t0\, \sqrt{1 + t0^2}}}{t0},\ dt \to \frac{dx0 + \sqrt{dx0^2 + dx0^2\, t0^2 - t0\, \sqrt{1 + t0^2}}}{t0}\right\}$$

In[ ]:=

```
t =.; x =.; λ =.; dt =.;
x0 = 0.5; t0 = -1.0; λmax = 2.65;
dx0 = 0.2; dt0 = dt /. initt;
Print["Initial Conditions:   x₀= ", x0, " t₀= ", t0, " ẋ₀= ", dx0, " ṫ₀= ", dt0]
dsol = NDSolve[{
    t''[λ] == -1/(2 (1 + t[λ]^2)) (-  t[λ]  t'[λ]^2 + 2   t'[λ] x'[λ] + t[λ] x'[λ]^2),
    x''[λ] == 1/(2 (1 + t[λ]^2)) ( (1 + 2 t[λ]^2) t'[λ]^2 - 2 t[λ] t'[λ] x'[λ] +   x'[λ]^2),
    t'[0]  == dt0,
    x'[0]  == dx0,
    t [0]  == t0  ,
    x [0]  == x0
   }, {t, x}, {λ, 0, λmax}
  ];
{ParametricPlot[Evaluate[{Mod[x[λ], 1.0], t[λ]} /. dsol],
  {λ, 0, λmax}, Exclusions → "Discontinuities", AxesLabel → {"x", "t"}],
 ParametricPlot3D[Evaluate[{Cos[2 π x[λ]], Sin[2 π x[λ]], t[λ]} /. dsol],
  {λ, 0, λmax}, AxesLabel → {"", "", "t"}],
 Plot[Evaluate[(-t[λ] t'[λ]^2/Sqrt[1 + t[λ]^2] + 2 t'[λ] x'[λ]/Sqrt[1 + t[λ]^2] + t[λ] x'[λ]^2/Sqrt[1 + t[λ]^2] - 1.0) /. dsol],
  {λ, 0, λmax}, AxesLabel → {"λ", "v^μ v_μ+1"}]}
```

Initial Conditions: x_0= 0.5 t_0= -1. $\dot{x}_0$= 0.2 $\dot{t}_0$= 1.02238

Out[ ]=



In[∘]:=
```
t=.; x=.; λ=.; dt=.;
x0 = 0.5; t0 = 0.8; λmax = 0.9;
dx0 = 0.8; dt0 = dt /. initt;
Print["Initial Conditions:  x₀= ", x0, " t₀= ", t0, " ẋ₀= ", dx0, " ṫ₀= ", dt0]
dsol = NDSolve[{
    t''[λ] == -1/(2 (1 + t[λ]^2)) (-     t[λ]  t'[λ]^2 + 2     t'[λ] x'[λ] + t[λ] x'[λ]^2),
    x''[λ] == 1/(2 (1 + t[λ]^2)) ( (1 + 2 t[λ]^2) t'[λ]^2 - 2 t[λ] t'[λ] x'[λ] +     x'[λ]^2),
    t'[0]  == dt0,
    x'[0]  == dx0,
    t [0]  == t0  ,
    x [0]  == x0
   }, {t, x}, {λ, 0, λmax}
  ];
{ParametricPlot[Evaluate[{Mod[x[λ], 1.0], t[λ]} /. dsol],
  {λ, 0, λmax}, Exclusions → "Discontinuities"],
 ParametricPlot3D[Evaluate[{Cos[2 π x[λ]], Sin[2 π x[λ]], t[λ]} /. dsol], {λ, 0, λmax}],
 Plot[Evaluate[(-t[λ] t'[λ]^2 + 2 t'[λ] x'[λ] + t[λ] x'[λ]^2 - Sqrt[1 + t[λ]^2]) /. dsol],
  {λ, 0, λmax}, AxesLabel → {"λ", "v^μ v_μ+1"}]}
```

Initial Conditions: x_0= 0.5 t_0= 0.8 $\dot{x}_0$= 0.8 $\dot{t}_0$= 0.801962

Out[∘]=



Misc

Compute length of curve: $x(\lambda) = \lambda \quad t(\lambda) = \lambda + t_0, \quad 0 \le \lambda < 1$

$$ds^2 = \frac{(\lambda+t_0)}{\sqrt{1+(\lambda+t_0)^2}}\, d\lambda^2 - \frac{2}{\sqrt{1+(\lambda+t_0)^2}}\, d\lambda^2 - \frac{(\lambda+t_0)}{\sqrt{1+(\lambda+t_0)^2}}\, d\lambda^2 = -\frac{2}{\sqrt{1+(\lambda+t_0)^2}}\, d\lambda^2 \Rightarrow$$

$$s = \int_0^1 |\, ds\, | = \int_0^1 \frac{\sqrt{2}}{(1+(\lambda+t_0)^2)^{1/4}}\, d\lambda$$

$$= \sqrt{2}\, (\lambda + t_0)\, {}_2F_1\left(\tfrac{1}{4}, \tfrac{1}{2}; \tfrac{3}{2}, -(t_0+\lambda)^2\right) \Big|_0^1 = \sqrt{2}\, \left\{-t_0\, {}_2F_1\left(\tfrac{1}{4}, \tfrac{1}{2}; \tfrac{3}{2}, -t_0{}^2\right) + (1+t_0)\ {}_2F_1\left(\tfrac{1}{4}, \tfrac{1}{2}; \tfrac{3}{2}, -(1+t_0)^2\right)\right\}$$

In[ ]:=
```
Integrate[√2/(1 + (λ + t0)^2)^(1/4), λ, Assumptions → t0 > 0]
```

Out[ ]=
$$\sqrt{2}\,(t0+\lambda)\,\text{Hypergeometric2F1}\left[\frac{1}{4},\frac{1}{2},\frac{3}{2},-(t0+\lambda)^2\right]$$

In[ ]:=
```
f[t0_] = Integrate[√2/(1 + (λ + t0)^2)^(1/4), {λ, 0, 1}, Assumptions → t0 > 0]
```

Out[ ]=
$$\sqrt{2}\left(-t0\,\text{Hypergeometric2F1}\left[\frac{1}{4},\frac{1}{2},\frac{3}{2},-t0^2\right]+(1+t0)\,\text{Hypergeometric2F1}\left[\frac{1}{4},\frac{1}{2},\frac{3}{2},-(1+t0)^2\right]\right)$$

In[ ]:=
```
Plot[f[t], {t, 0, 10}]
```

Out[ ]=


Orthogonal basis:

In[ ]:=
```
gmatrix = 1/Sqrt[1 + t^2] ( t  -1
                            -1 -t ); gmatrixinv = Inverse[gmatrix] // Simplify;
evecs = Eigenvectors[gmatrix]; evecsinverse = Inverse[evecs] // Simplify;
Print[
 "g^-1= ", gmatrixinv // MatrixForm,
 "       g - g^-1= ", gmatrix - gmatrixinv // MatrixForm,
 "\ngD = ", DiagonalMatrix[Eigenvalues[gmatrix]] // MatrixForm,
 "\nO  = ", evecs // MatrixForm,
 "\ngD = ", evecs . gmatrix . evecsinverse // Simplify // MatrixForm,
 "\ngD = ", evecs . gmatrix .Transpose[evecs] // Simplify // MatrixForm,
 "    (diagonal, but not ±1)",
 "\nOrthogonality of vectors:",
 "\n e0= ", evecs[[1]], "     e1= ", evecs[[2]], "  (not orthormal)",
 "\n e0·e1= ", evecs[[1]].gmatrix.evecs[[2]] // Simplify,
 "\n e0·e0= ", evecs[[1]].gmatrix.evecs[[1]] // Simplify,
 "\n e1·e1= ", evecs[[2]].gmatrix.evecs[[2]] // Simplify
]
```

$g^{-1}= \begin{pmatrix} \frac{t}{\sqrt{1+t^2}} & -\frac{1}{\sqrt{1+t^2}} \\ -\frac{1}{\sqrt{1+t^2}} & -\frac{t}{\sqrt{1+t^2}} \end{pmatrix}$ $\quad g - g^{-1}= \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$

$g_D = \begin{pmatrix} -1 & 0 \\ 0 & 1 \end{pmatrix}$

$O = \begin{pmatrix} -t+\sqrt{1+t^2} & 1 \\ -t-\sqrt{1+t^2} & 1 \end{pmatrix}$

$g_D = \begin{pmatrix} -1 & 0 \\ 0 & 1 \end{pmatrix}$

$g_D = \begin{pmatrix} -2-2t^2+2t\sqrt{1+t^2} & 0 \\ 0 & 2\left(1+t^2+t\sqrt{1+t^2}\right) \end{pmatrix}$ (diagonal, but not ±1)

Orthogonality of vectors:

$e_0= \{-t+\sqrt{1+t^2}, 1\}$ $\quad e_1= \{-t-\sqrt{1+t^2}, 1\}$ (not orthormal)

$e_0 \cdot e_1= 0$

$e_0 \cdot e_0= -2-2t^2+2t\sqrt{1+t^2}$

$e_1 \cdot e_1= 2\left(1+t^2+t\sqrt{1+t^2}\right)$

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.

Email: konstant@mail.ntua.gr
Web: http://physics.ntua.gr/konstant